SUMMARY = "Ethernet bridge frame table administration"
DESCRIPTION = "Ebtables is used to set up, maintain, and inspect the tables of \
 Ethernet frame rules in the Linux kernel. It is analogous to iptables, \
 but operates at the MAC layer rather than the IP layer."
HOMEPAGE = "http://ebtables.sourceforge.net"

PR = "r0"
inherit debian-package

LICENSE = "GPLv2"
LIC_FILES_CHKSUM = "file://COPYING;md5=53b4a999993871a28ab1488fdbd2e73e"

inherit autotools-brokensep
CFLAGS += "-fstack-protector-all"
EXTRA_OEMAKE = " \
	'LIBDIR=${base_libdir}/${PN}' \
	'BINDIR=${base_sbindir}' \
	'CC=${CC}' \
	'CFLAGS=${CFLAGS}' \
	'LDFLAGS=${LDFLAGS}' \
	"
do_install_append() {
	# base on debian/rules
	mv ${D}${sysconfdir}/default/ebtables-config \
	   ${D}${sysconfdir}/default/ebtables
	rm -rf ${D}${base_sbindir}/ebtables-save \
	       ${D}${base_sbindir}/ebtables-restore
}
RRECOMMENDS_${PN} += "iptables kmod"
# Avoid a parallel build problem
PARALLEL_MAKE = ""
FILES_${PN} += "${base_libdir}/${PN}/*"
FILES_${PN}-dbg += "${base_libdir}/${PN}/.debug"

Add recipe for ebtables package
ebtables: Add recipe for ebtables package

Create recipe to build ebtables package from Debian's source code

Signed-off-by: Nguyen Van Hieu <95b6c63292b73e14e1eaecc80ea875350fcf3ae9@toshiba-tsdv.com>
Signed-off-by: Masahiro Yamada <76d6c03d331b29d77eaac66c3113243c4dd22703@toshiba.co.jp>
